

# Vážení zákazníci,

dovolujeme si Vás upozornit, že na tuto ukázkou knihy se vztahují autorská práva, tzv. copyright.

To znamená, že ukáзка má sloužit výhradně pro osobní potřebu potenciálního kupujícího (aby čtenář viděl, jakým způsobem je titul zpracován a mohl se také podle tohoto, jako jednoho z parametrů, rozhodnout, zda titul koupí či ne).

Z toho vyplývá, že není dovoleno tuto ukázkou jakýmkoliv způsobem dále šířit, veřejně či neveřejně např. umístováním na datová média, na jiné internetové stránky (ani prostřednictvím odkazů) apod.

*redakce nakladatelství BEN – technická literatura*  
[redakce@ben.cz](mailto:redakce@ben.cz)



# 6 UNIVERZÁLNÍ ROZHRAŇÍ USB

V této kapitole představíme konkrétní vývojovou aplikaci USB. Úkolem je existující rozhraní na RS232 (CompuLAB firmy Modul-Bus) nově vyvinout pro USB. Nejdůležitější kroky tohoto vývoje jsou zde dokumentovány a mohou sloužit jako základ pro vlastní pokusy. Hotové zařízení je vyráběno firmou Modul-BUS pod označením CompuLAB-USB. Také vlastní stavba je možná, protože jsou k dispozici všechny potřebné informace.

Technická data starého rozhraní CompuLAB mají být zachována nebo zlepšena. Z toho vyplývají následující minimální požadavky na nově vyvinuté rozhraní:

- 8 digitálních výstupů, kompatibilních s TTL,
- 8 digitálních vstupů, kompatibilních s CMOS, s vysokým vstupním odporem,
- 2 analogové vstupy, rozlišení 8 bitů, rozsah 0 až 5 V.

Protože obvod CY7C63000 na rozdíl např. od řadiče svého předchůdce ST6 nemá žádný interní A/D převodník, musí se nejprve vyhledat vhodný A/D převodník. Není-li vhodný typ je TLC541 s 11 vstupy a rozlišovací schopností 8 bitů. Proti tomuto převodníku však hovoří to, že vyžaduje externí hodinový signál 2 MHz. Nejbližší lepší převodník od Texas Instruments je TLC1543 s rozlišením 10 bitů. Má interní generátor hodin a vyžaduje pro vybuzení čtyři nebo pět linek portu, podle toho, je-li vyhodnocován signál End-Of-Conversion, případně se použije jednoduchá časová smyčka. Na tento převodník padla volba také proto, že má příznivý poměr cena/výkon.


## 6.1 A/D PŘEVODNÍK TLC1543

A/D převodník TLC1543 je integrovaný obvod CMOS se sériově taktovaným rozhraním k přímému buzení mikrořadičem. Digitální data se tedy objevují na jediné lince a jsou mikrořadičem načítána postupně. Linka hodinového signálu řídí přenos dat jako u posuvného registru. Stejným způsobem se data, zde speciálně adresa vstupního kanálu, přenáší z mikrořadiče do převodníku.

Napájecí napětí je 5 V, typický proudový odběr je asi 1 mA. Převodník pracuje metodou postupných aproximací se sítí z binárně vážených kondenzátorů. Obr. 6.1 ukazuje principiální schéma.

Komparátor CMOS testuje každý bit porovnáním náboje. V první fázi převodu se převezme analogová hodnota sepnutím spínače S-C a všech spínačů S-T. Všechny kondenzátory se přitom nabijí na vstupní napětí, přičemž jejich společný přívod je na potenciálu spínacího prahu komparátoru rovného asi polovině napájecího napětí. Potom se spínače rozepnou. Komparátor pak testuje napětí každého kondenzátoru. Nejprve se připojí uzel 512 (node 512) na referenční napětí REF+ a všechny ostatní uzly na REF-. Když výsledné napětí leží pod prahovým napě-


tím, bude pro bit 9 rozpoznána jednička a kondenzátor zůstane na REF+. Tentýž proces se pak bude opakovat s kondenzátorem s váhou 256, aby se získal bit 8. Při každém kroku postupné aproximace se původní náboj rozdělí na všechny kondenzátory. Po celkem deseti porovnáních bude napětí určeno s přesností na 10 bitů.




Obr. 6.1 Princip převodníku se spinanými kondenzátory (Texas Instruments)

Vstup převodníku je interním multiplexerem přiveden vždy na jeden z 11 externích vstupů. Navíc existují tři interní kanály, které umožňují autotest. Kanál A11 je interně přiveden na polovinu referenčního napětí a dává výsledek 512. Možné odchylky poskytují cenné informace o případném rušení, např. nedostatečném vyfiltrování napájecího napětí.

Obvod TLC1543 používá k aktivaci přenosu dat s radičem linku *chip-select* ( $\overline{CS}$ ).




Obr. 6.2 Blokové schéma obvodu TLC1543



Obr. 6.3 Rozložení vývodů obvodu TLC1543


Převod A/D se provádí, pokud je  $\overline{CS}$  na vysoké úrovni, tedy neaktivní. Každý taktovaný přenos dat dodává současně 10 datových bitů posledního převodu mikrořadiče a na A/D převodník posílá adresu následujícího převodu.

Převodník jsme pro první pokus chtěli připojit na desku vývojového kitu s co nejmenšími změnami softwaru. Nabízí se nahradit rutinu teploměru rutinou AD. Odpovídajícím způsobem se použijí vývody portů P0.0 až P0.4.



Obr. 6.4 Graf průběhu impulzů při vybuzení

Přímá náhrada rutiny teploměru beze změn na jádru programu řadiče vede k tomu, že AD rutina TLC1543a.asm řízená timerem je vyvolána každých 1024  $\mu$ s. Výsledky převodu se ukládají do paměťových buněk 78h a 79h.



Obr. 6.5 Připojení převodníku na port 0

```

; TLC1543a.asm - 10 Bit 12 Chanell AD-Converter
; Anschluss an Port 0

ADPort          :equ 00h          ; SysPort0
ADMaskBits      :equ 1fh          ;
ADEOC           :equ 01h          ; P0.0
ADClock         :equ 02h          ; P0.1
ADDataIn        :equ 04h          ; P0.2
ADDataOut       :equ 08h          ; P0.3
ADCS            :equ 10h          ; P0.4

gbADChan        :equ 31h          ;
gbADPortMirror  :equ 32h          ;
gbADBitcount    :equ 33h          ;
gbADChanIn      :equ 34h          ;
gbADDatHi       :equ 78h          ;USBEndP1FIFO
gbADDatLo       :equ 79h          ;USBEndP1FIFO +1

; //$PAGE
;*****
; ADRead()
;
;*****

ADRead:
    mov a,[gbADChanIn]            ;read analog input
    mov [gbADChan],a
    mov a,[gbPort0]              ; setup port bits
    and a,~ADClock
    or a,(ADDataOut|ADCS|ADEOC)
    iowr ADPort
    mov [gbADPortMirror],a
    mov a,00h
    mov [gbADDatHi],a
    mov [gbADDatLo],a
    mov a,[gbADPortMirror]        ;CS low
    and a,~ADCS
    iowr ADPort
    mov [gbADPortMirror],a
    mov a,8                        ;8 bits count
ADLoop1:
    mov [gbADBitcount],a

```

```

mov a, [gbADDatHi] ;output data shift left
asl a
mov [gbADDatHi], a
iord ADPort ;read one Bit from Dout
and a, ADDDataOut
cmp a, ADDDataOut
jnz Ainlow1
mov a, [gbADDatHi]
or a, 01h
mov [gbADDatHi], a
Ainlow1:
mov a, [gbADPortMirror] ;output one bit to Din
and a, ~ADDDataIn
mov [gbADPortMirror], a
mov a, [gbADChan]
and a, 80h
cmp a, 80h
jnz AoutLow
mov a, [gbADPortMirror]
or a, ADDDataIn
mov [gbADPortMirror], a
AoutLow:
mov a, [gbADPortMirror]
iowr ADPort
or a, ADClock ; set clock high
iowr ADPort
mov a, [gbADChan] ;input data shift left
asl a
mov [gbADChan], a
nop
nop
nop
nop
mov a, [gbADPortMirror]
and a, ~ADClock ; set clock low
iowr ADPort
mov [gbADPortMirror], a
mov a, [gbADBitcount] ; complete loop 1
dec a
jnz ADLoop1
; read the two bytes left

```

```

mov a,02h
ADLoop2:
mov [gbADBitcount],a
mov a,[gbADDatLo]                ;output data shift
    left

asl a
mov [gbADDatLo],a
iord ADPort                      ;read one Bit from
    Dout

and a,ADDataOut
cmp a,ADDataOut
jnz Ainlow2
mov a,[gbADDatLo]
or a,01h
mov [gbADDatLo],a
Ainlow2:
mov a,[gbADPortMirror]          ;set clock high
or a,ADClock
iowr ADPort
nop
nop
nop
nop
and a,~ADClock                  ;set clock low
iowr ADPort
mov [gbADPortMirror],a
mov a,[gbADBitcount]            ;complete loop 2
dec a
jnz ADLoop2
mov a,[gbADPortMirror]          ;set CS high
or a,ADCS
iowr ADPort
ret

```

*Listing 6.1 Buzení TLC1543 prostřednictvím portu 0.*

Program používá dvě smyčky s celkem 12 hodinovými impulzy. V první smyčce se čte 12 datových bitů a 8 datových bitů se posune do převodníku. Hlavní program předá v globálním bytu gbADChanIn příslušný analogový vstup. Protože první čtyři přenášené bity představují adresu pro interní multiplexer A/D převodníku, přenášejí se vždy násobky 10h (tj. 16) (00, 16, 32, 48 ...). Rutina okopíruje číslo kanálu do proměnné gbADChan, která se ve smyčce osmkrát posune doleva, aby

se izoloval vždy jeden bit. Podobným způsobem se postupně načítá prvních osm bitů posledního převodu a posouvá se do gbADDatHi.

Druhá smyčka zpracovává u 10bitového měření oba bity převodu s nejnižší váhou. Zde již není potřebný žádný výstup na převodník. Nižší byte (*lowbyte*) se uloží na adresu 79h v nepoužité paměti FIFO koncového bodu 1. Uživatel může v závislosti na požadavcích číst jen vyšší byte (*highbyte*) na adrese 78h a získat tak přesnost 8 bitů nebo použije i *lowbyte*, aby se přesnost zvýšila na 10 bitů. Celá rutina AD potřebuje asi 140  $\mu$ s. Není nutné brát ohled na dobu konverze, protože převod se spouští jen jednou za milisekundu. Tím je k dispozici 860  $\mu$ s, zatímco převodník vystačí s 21  $\mu$ s. Na druhé straně byly do smyčky zařazeny instrukce NOP (no operation), aby byla zajištěna dostatečně dlouhá fáze High hodinových impulzů. Poměrně velké zvyšovací odpory a velmi malý výstupní proud na portu 1 vedly spolu s nevyhnutelnými parazitními kapacitami k zaoblení hodinových impulzů, což se ukázalo tolerovatelné jen při prodloužení hodinových impulzů.

V tomto prvním pokusu byla data připravena přímo ve výstupní oblasti 78/79h. Ukázalo se však, že jistý procentuální podíl dat byl narušen. Jako příčina přicházelo v úvahu jen zpracování přerušení. AD rutina samotná běží s přerušením od časovače 1024  $\mu$ s. Může být přerušena, dojde-li k události koncového bodu 0. AD rutina potřebuje asi 140  $\mu$ s. Je tedy pravděpodobnost asi 14 %, že proběhne přístup pro čtení výsledku, když ještě běží převod.

Zlepšení se dosáhne, jestliže se data připraví na dvou jiných paměťových místech a teprve nakonec se překopírují na výstup.

Jiným problémem je časové chování. Řídící dotaz (*Control request*) vyžaduje vždy tři USB rámy, tedy 3 ms. Na každý kanál je zapotřebí jeden zápisový přístup a dva čtecí přístupy na RAM. Tím dojdeme k časové potřebě 9 milisekund na kanál. Naproti tomu u RS232, předchůdce USB, je zapotřebí jen jedna milisekunda. Zde byl jen jeden dotazovací příkaz na kanál. Navíc byl vyslán byte a přijat byte odpovědi. Při přenosové rychlosti 19 200 baudů z toho vyplývala doba asi 1 ms.


## 6.2 SHRNUTÍ KANÁLŮ AD

Nevýhodu pomalejšího dotazování na zařízení USB je možno vykompenzovat tím, že se současně načítá několik kanálů. Jedním dotazem je možno společně přenést až 8 bytů. Díky tomu je možno dotazovat se současně na čtyři kanály. Přenos čísla kanálu není nutný, jestliže firmware automaticky inkrementuje číslo kanálu a provádí několik měření dohromady. Vlastní měření na čtyřech kanálech může probíhat v přerušení od timeru, potřebuje-li výrazně méně než 1 milisekundu. Měřená data pro čtyři kanály se obnovují jednou za milisekundu. Dotaz na čtyři kanály vyžaduje jen tři milisekundy, je-li k dispozici odpovídající funkce ovladače pro dotaz na osm bytů. Tím se získá malá výhoda v rychlosti oproti RS232.

Konečné využití portu by mělo vystačit s minimem součástek. Protože A/D převodník s 11 vstupními kanály nabízí víc, než je zapotřebí, je nasnadě myšlenka využít 8 analogových vstupů jako digitální vstupy. To vyžaduje jedno měření na každém z osmi kanálů a vážení výsledků měření. V konečném výsledku je tímto způsobem možno využívat port 0 mikrořadiče zcela pro digitální výstupy, kdežto

port 1 se svými čtyřmi linkami obsluhuje A/D převodník. Linka EOC se nevyhodnocuje, takže čtyři linky portu stačí. Při vhodném uspořádání je možno budít i LED pro indikaci připravenosti.

Zapojení podle obr. 6.6 používá prvých osm vstupních kanálů jako digitální vstupy. Tím jsou k dispozici tři vstupy jako analogové. Dodatečný vstup představuje zlepšení oproti původnímu rozhraní (CompuLAB = RSS232). Při vhodném dimenzování vstupních obvodů je nadto možno dosáhnout i toho, že všechny digitální vstupy lze volitelně využívat také jako analogové vstupy. V případě potřeby bychom tedy měli všech 11 analogových vstupů.



Obr. 6.6 Přirazení portů

Z časových důvodů se mají čtyři AD kanály měřit společně. Uživatel může určit počáteční kanál, od něhož se budou kanály počítat. Rutina převodníku se k tomu účelu musí vyvolat pětkrát, protože převedená data se mohou načítat vždy teprve při následujícím volání. Mezi jednotlivými sériovými přístupy na převodník se tentokrát musí dodržovat doba převodu 21  $\mu$ s. Příkazy NOP pro prodloužení hodinových impulzů si však můžeme ušetřit, protože výkonnější výstupy portu 1 dávají strmější hrany impulzů. Celkově je převod proveden za 650  $\mu$ s, což je ještě dostatečně daleko od mezní hodnoty 1024  $\mu$ s. Zbýlých 350  $\mu$ s musí stačit pro jiné události vyvolávající přerušení, zejména pro dotazy (*requests*) USB.

```
; TLC1543b.asm - 10 Bit 12 Kanal AD-Converter
```

```
ADPort          :equ 01h          ; SysPort1
ADMASKBITS      :equ 1fh          ;
ADClock         :equ 01h          ; P1.0
ADDataIn        :equ 02h          ; P1.1
ADDataOut       :equ 04h          ; P1.2
ADCS            :equ 08h          ; P1.3
```